Program Title:	SKI JUMP

Let's attempt the longest jump!

Today there is fine weather, ideal for ski-jumping. You are enthusiastic about the set-up of the longest jump distance. Jump with perfect timing after level skiing, keeping your balance despite wind from the right and left sides.

Now, let us see how many meters you can make!

■ HOW TO PLAY

1. [R] [U] [N] [ENTER] displays the title "The Longest Distance". Press the space key and then the game starts.
2. After the jumping stand is displayed on the screen, the man starts skiing. During the course of his level skiing, press the key [5] to let him jump.
   If the jumper fails to jump due to a failure of timing, "0m" will be displayed on the screen and then the game returns to the initial display of the title.
3. Although the skier has jumped, keep the jumper's balance despite wind from the left and right sides by using the following keys.

   Wind from left side (→): Press key [4]
   Wind from right side (←): Press key [6]

   (The jumper's balance is indicated by the display of a vertical bar "I".)

   - When the jumper lands properly on the ground, the distance covered is displayed, but when he falls halfway due to losing his balance, "Fallen" is displayed on the screen.
     In such a case, no display of the flying distance will appear on the screen.

     When the above procedure is over, the initial display of the game will reappear. Now, you should wait to input the game if you wish to try again.

■ REFERENCE

⟨The method of calculating the distance covered⟩

1. Timing of the jump
   J (the descending speed of the jumper depending on the timing) is calculated through the X coordinate (125 – 110) when pressing key [5].

$$J = (X - 110) \times 0.02$$

Move the jumper along each dot up to X-coordinates 105 – 80. As in every movement, the jumper decreases in accordance with the formula of "Y = Y + J" (Y is for the altitude).
In short, the timing of pressing key 5 affects the altitude at the point of the X-coordinates 80.

2. Balance
The altitude decreases at a regular speed. The X-coordinates are calculated by the following formula.

$$X = X - (5 - ABS(B)) \times 0.2$$

B shows the balance and covers the range of the figures from –5 to +5. (When B is above or below the figure, the jumper falls down.)

Calculation and display are repeated with the above sequence and then S (flying distance) is calculated from the X-coordinates when the landing judgement formula "Y > 31 – INT (X x 0.05)" is established.

$$S = (100 - X) \times 1.5$$

Therefore, keeping the proper balance is the knack of increasing the distance covered in flight.

■ KEY OPERATION SEQUENCE

1. R U N ENTER [Program starts at]

```
*** Ski Jump ***
The Longest Distance:
                                   Øm
Press the space key
```

2. [The game starts]



3. 5



4. 4



Continue the game by operating 4 or 6 key.

《display example of the well landing》



《display example in the case of falling halfway》



■ PROGRAM LIST

```
10:WAIT 0:H=0
20:F1$="40888890F08000":F2$="40888890F
   0800000":F3$="1EE05E201824"
30:DIM S$(5)*16:S$(1)="030C30C00000000
   0":S$(2)="00000FF000000000"
40:S$(3)="000000FF000000":S$(4)="00000
   0F00F000000":S$(5)="0000000C0300C0
   3"
50:CLS : WAIT 0: CURSOR 1,0: PRINT "**
   * Ski Jump ***": BEEP 1
60:CURSOR 0,1: PRINT "The Longest Dist
   ance:";
70:CURSOR 17,2: PRINT USING "####";H;"
   m";
80:CURSOR 0,3: PRINT "Press the space
   key";
90:I$= INKEY$ : IF I$<>" " THEN 90
100:CLS : LINE (149,5)-(140,5): LINE -(
   130,15): LINE -(110,15): LINE -(99,
   26): GOTO 120
110:FOR I=2 TO 5: LINE (119-I*20,26+I)-
   (100-I*20,26+I): NEXT I: RETURN
120:FOR I=1 TO 5: LINE (119-I*20,26+I)-
   (100-I*20,26+I): NEXT I
130:GCURSOR (141,4): GPRINT F1$:
   GCURSOR (141,4): BEEP 3: GPRINT "00
   0000000000"
140:FOR I=0 TO 10: GCURSOR (134-I,4+I):
   GPRINT F1$: NEXT I
150:X=125
160:I$= INKEY$ : IF I$="5" THEN 220
170:X=X-2: GCURSOR (X,14): GPRINT F2$:
   IF X>110 THEN 160
180:GCURSOR (X,14): GPRINT "0000000000
   0"
190:FOR I=0 TO 11: GCURSOR (103-I,15+I)
   : GPRINT F1$: NEXT I
200:CURSOR 19,3: WAIT 160: PRINT "0 m";
210:GOTO 50
220:J=(X-110)*.02: FOR I=X TO 105 STEP
   -2: GCURSOR (I,14): GPRINT F2$:
   NEXT I
230:Y=14: FOR I=105 TO 80 STEP -1:Y=Y+J
   : GCURSOR (I,Y): GPRINT F1$: NEXT I
   :B=0,X=80
240:R= RND 3: ON R GOTO 250,260,270
250:GCURSOR (0,10): GPRINT "1010101092
   543810":B=B+1: GOTO 280
260:GCURSOR (0,10): GPRINT "0000000000
   00000": GOTO 280
270:GCURSOR (0,10): GPRINT "1038549210
   101010":B=B-1
280:I$= INKEY$
290:IF I$="4" LET B=B-1
300:IF I$="6" LET B=B+1
310:IF B<-5 OR B>5 THEN 400
320:GCURSOR (50,10): GPRINT S$(B*.5+3)
330:X=X-(5- ABS (B))*.2:Y=Y+.26: IF Y>3
   1- INT (X*.05) THEN 370
340:GCURSOR (X,Y): GPRINT F2$
350:IF Y>27 LET C= POINT (X,Y+1): IF C
   GOSUB 110
360:GOTO 240
370:S=(100-X)*1.5: CURSOR 18,3: PRINT
   USING "####";S;: WAIT 320: PRINT "m
   "
380:IF S>H LET H=S
390:GOTO 50
400:FOR I=Y TO 31: GCURSOR (X,I):
   GPRINT F3$: NEXT I
410:CURSOR 17,3: WAIT 160: PRINT "Falle
   n";
420:GOTO 50
```

1279 bytes

■ MEMORY CONTENTS

B	balance
C	√
H	The longest flying distance
I, I$	loop counter, √
J	descending speed
R	direction of wind
S	flying distance
X	level distance
Y	altitude
F1$	for display (figure for man)
F2$	for display (figure for man)
F3$	for display (figure for man)
S$(5)	for display (balance)